APL SYMBOLIC MANIPULATION AND GENERATING FUNCTIONS

A. Kellerman
IBM Corporation
System Products Division
P. O. Box 6
Endicott, New York 13760

ABSTRACT

This paper is a continuation of the work of Kellerman and Rodgers [1], who covered some basic techniques of solving combinatorics problems using several features of the APL language. This paper presents a powerful additional technique: the use of generating functions.

INTRODUCTION

In the solution of combinatorics problems, APL is often appropriately used to evaluate functions that have been defined for specific values of the function arguments. In using generating functions, the modus operandi is such that the coefficients of the generating functions themselves contain the desired solutions to the combinatorial problems of interest. The coefficients are usually obtained after functions are multiplied, added, or divided symbolically. Although the necessary APL primitive functions do not exist for symbolic manipulation, a set of "primitive" APL functions has been developed [2] that can handle the necessary symbolic manipulation that is needed.

As an example of a generating function consider the one that represents the possible combinations of 3 objects a, b, and c. The generating function for

$$\underbrace{1}_{\text{0 out of 3}} + \underbrace{(a+b+c)T}_{\text{1 out of 3}} + \underbrace{(ab+ac+bc)T^2}_{\text{2 out of 3}} + \underbrace{(abc)T^3}_{\text{3 out of 3}}$$

is:

$$C_3(T) = (1+aT)(1+bT)(1+cT)$$

or generalizing for n objects:

$$C_n(T) = (1+a_1T)(1+a_2T)\ldots(1+a_nT)$$
$$= 1+b_1T + b_2T^2+\ldots+ b_nT^n$$

where the coefficients (b_i) contain, after multiplication, an exhibit of the combinations.

PARTITIONS

Let a, b, c, d, e, and f be unequal positive integers. Then the coefficient of x^n in the expansion of

$$(1+x^a+x^{2a}+\ldots)(1+x^b+x^{2b}+\ldots)\ldots$$
$$(1+x^f+x^{2f}+\ldots) \qquad (1)$$

equals the number of partitions of n with summands restricted to a, b, c, d, e, and f. In each factor terms must be included up to all coefficients not exceeding n.

Example: $(1+x^2+x^4+x^6+x^8+x^{10}+x^{12})\cdot$

$(1+x^4+x^8+x^{12})(1+x^6+x^{12})(1+x^8)(1+x^{10})(1+x^{12})$

will yield as the coefficient of x^{12} the number of ways to sum up to 12 with even numbers. Here, a is 2, b is 4, c is 6, d is 8, e is 10, and f is 12.

A polynomial can be represented symbolically in APL by a two-row matrix whose first row contains the exponents and whose second row contains the coefficients. Functions for symbolic operations such as multiplication, addition, subtraction, and division on such polynomials (described in [2]) are listed in the Appendix.

Using the function for multiplication, the number of partitions of N with distinct summands in VC can be obtained from the function PARTITION as shown below:

```
      ∇PARTITION[⎕]∇
    ∇ R←VC PARTITION N
[1]   I←ρVC←VC[⍋VC]
[2]   R←(0,(R[1;]⍳N)-1)↓R←((((0,⍳
      ⌊N÷VC[I-1])×VC[I-1])),[0.5] 1
      ) M((((0,⍳⌊N÷VC[I])×VC[I])),
      [0.5] 1)
[3]   I←I-1
[4]   REP:→((I←I-1)=0)/END
[5]   R←(0,(R[1;]⍳N)-1)↓R←(((,⍳⌊N÷VC
      [I])×VC[I]),[0.5] 1) M R
[6]   →REP
[7]   END:R←R[2;(R[1;]⍳N)]
    ∇
```

Example 2.1:
```
      5 10 15 PARTITION 20
4
```
different ways to add to 20 using 5, 10 and 15.

Example 2.2: `1 5 10 25 50 100 PARTITION 100`

`92` different ways to partition a dollar (make change).

Example 2.3: In Reference [1], a function SOLVE was developed to list the solutions of $X_1 + X_2 + X_3 = 6$. What if only the number of solutions is desired? Another interpretation of the coefficient of X^n in Eq. (1) is that of giving the number of solutions to $aX_1 + bX_2 + cX_3 + dX_4 + eX_5 + fX_6 = n$. For each solution there corresponds a partition of n with summands a, b, c, d, e, and f. This idea can be extended to any number of unknowns X_j with accompanying coefficients a, b, c, d, e, f, g, h, ...

Example 2.4: How many solutions are there of $X_1 + X_2 + X_3 = 6$? Look at the coefficient of X^6 in Eq. 1, i.e.:

$(1 + \ldots + X^6)(1 \ldots + X^6)(1 + \ldots + X^6)$. Using A M B M C, where A, B, and C represent $(1 + \ldots + X^6)$, the coefficient is determined to be 10. If desired, we can tabulate these solutions using L SOLVE M from [1]:

```
      3 SOLVE 6

   114
   123
   132
   141
   213
   222
   231
   312
   321
   411
```

POLYA'S COUNTING THEORY

A dominant position in combinatorics is held by Polya who combined the aspects of using generating functions, counting equivalence classes (rather than the objects themselves), and using weights, into a fundamental theorem. His theory enables one to answer in a completely systematic and complete way such questions as:

> Given a cube with faces of two colors, red and blue, how many distinct patterns show four red faces and two blue faces?

To consider another somewhat more practical problem:

> In determining the routing for the interconnections of digital circuitry it is necessary to examine several different placements of the building units (modules) on cards. How many "different" placements are there?

This is the combinatorial problem of determining the number of equivalence classes of module-on-card placements. Polya's theorem says that if we wish to determine the number of equivalence classes (i.e., the number of inequivalent configurations), given a figure counting series $F(x_1, x_2, \ldots, x_k)$, and a permutation group G with cycle index $Z_G(f_1, f_2, \ldots, f_s)$, the following theorem allows us to do this.

Polya's Theorem: The (inequivalent) configuration-counting series is

$$C(x_1, x_2, \ldots, x_k) = Z_G(F(x_1, \ldots, x_k),\ F(x_1^2, \ldots, x_k^2),\ \ldots,\ F(x_1^s, \ldots, x_k^s)^s\ )$$

where the coefficient of $x_1^{j_1} x_2^{j_2} \ldots x_k^{j_k}$ in C is the number of equivalence classes under G whose members have weight vector $(j_1, j_2, \ldots, j_k)$.

It is outside the scope of this paper to go into the details of this theorem. It is hoped that the examples will illustrate its usefulness in solving combinatorial problems. For further information, the reader is encouraged to consult References [3-5]. It might be pointed out here that not only have APL and APL graphics been applicable in calculating the final result, but in originally understanding what is really a very complex theory, APL has been used extensively. Also to generate more complex permutation groups, and check for closure, APL is also now being used. In the fairly simple examples given here, it is unnecessary to use APL in this way.

Example 2.5: Consider a 6-sided cube with 8 vertices (corners). Let G be the group of all the permutations of S that can be produced by rotations of the cube. In the three cases shown in Table I, the first S is the set of all faces, then the set of all vertices, and finally the set of all edges. The entries, coeff. $x_j^I y_k^1$, are the contributions to the cycle index from each category of rotation.

Table I

Tabulation of the Components of the Cycle Index for Rotations of a Cube

All possible rotations of the cube (24 total rotations)	S: set of all faces	S: set of all vertices	S: set of all edges
Identity	y_1^6	y_1^8	y_1^{12}
3 rotations of 180 degrees along lines connecting 2 faces	$3y_1^2y_2^2$	$3y_2^4$	$3y_2^6$
6 rotations of 90 degrees (3 clockwise, 3 counterclockwise)	$6y_1^2y_4^1$	$6y_4^2$	$6y_4^3$
6 rotations of 180 degrees along lines connecting opposite vertices	$6y_2^3$	$6y_2^4$	$6y_1^2y_2^5$
8 rotations of 120 degrees along lines connecting opposite vertices (4 cw, 4 ccw)	$8y_3^2$	$8y_1^2y_3^2$	$8y_3^4$

For example, consider one of the 3 rotations of 180 degrees along lines connecting 2 faces, as shown in Fig. 1 (note that S is the set of all vertices).



Fig. 1 — One of the Three Possible Rotations of a Cube along an Axis Connecting Faces

The entry shown in Fig. 1 is 4 cycles of length 2 (the length is the number of elements in each cycle) and thus the entry in Table I for 3 rotations is $3y_2^4$. The three rotations are about an axis through E and F, about an axis through B and D, and through C and A, as shown in Fig. 2.



Fig. 2 — Rotations of the Cube about Axis through Faces

The cycle indices for the sets S from Table I are as follows:

Faces

$$z_G = \frac{1}{24}(y_1^6 + 3y_1^2y_2^2 + 6y_1^2y_4 + 6y_2^3 + 8y_3^2) \qquad (2)$$

Edges

$$z_G = \frac{1}{24}(y_1^{12} + 3y_2^6 + 6y_4^3 + 6y_1^2y_2^5 + 8y_3^4)$$

Vertices

$$z_G = \frac{1}{24}(y_1^8 + 9y_2^4 + 6y_4^2 + 8y_1^2y_3^2)$$

Now, to answer the original question posed. That is, if the cube has either red or blue faces, how many distinct color patterns show 4 red faces and 2 blue faces. We can give weight x_1 to red and x_2 to blue. Then, the figure counting series F is $x_1 + x_2$ and the configuration counting series is obtained by substituting the following:

$$y_1 \leftrightarrow x_1 + x_2$$
$$y_2 \leftrightarrow x_1^2 + x_2^2$$
$$y_3 \leftrightarrow x_1^3 + x_2^3$$
$$y_4 \leftrightarrow x_1^4 + x_2^4$$

into the cycle index Z_G, for faces (Eq. 2) or:

$$C = \frac{1}{24}(x_1 + x_2)^6 + 3(x_1 + x_2)^2(x_1^2 + x_2^2)^2 +$$
$$6(x_1 + x_2)^2(x_1^4 + x_2^4) + 6(x_1^2 + x_2^2)^3 + 8(x_1^3 + x_2^3)^2$$

Multiplying out in APL, we obtain:

```
600    501    402    303    204    105    6
  1      1      2      2      2      1    1
```

$2x_1^4x_2^2$

where the top row contains the exponents, and the second row the coefficients.

The coefficient of $x_1^4x_2^2$ is 2. There are thus 2 color patterns. The APL technique for obtaining this result will be illustrated in detail in example 2.6.

Example 2.6: Let us consider a 2 by 3 rectangle in a 2-dimensional space on which we are to place objects with the following properties:

(a) Round and red
(b) Square and red
(c) Round and blue
(d) Square and blue.

Assign the following numbers to each of the possible properties: 1-red, 2-blue, 3-round, 4-square. In determining the figure-counting series, the following procedure is performed: if a figure has property i then place a one in the i-th position of the weight vector for that figure, otherwise place a zero. Thus the figure-counting series is:

$$F(x_1, x_2, x_3, x_4) = x_1x_3 + x_1x_4 + x_2x_3 + x_2x_4$$
$$= (x_1 + x_2)(x_3 + x_4)$$

Let G be the group of the permutations resulting from all possible rotations and reflections. The six numbered squares are the set S (see Table II).

Table II

Permutation Group for 2 by 3 Rectangle

Permutation	Resulting Configuration	Cycles	Type
(a) Identity	1 2 3 / 4 5 6	1↻ 2↻ 3↻ 4↻ 5↻ 6↻	y_1^6
(b) Right and left exchange	3 2 1 / 6 5 4	1⇄3 2↻ 4⇄6 5↻	$y_1^2y_2^2$
(c) Top and bottom exchange	4 5 6 / 1 2 3	1⇄4 2⇄5 3⇄6	y_2^3
(d) Permutation (b) followed by (c)	6 5 4 / 3 2 1	1⇄6 2⇄5 3⇄4	y_2^3

It is easy to verify that the permutations in Table II form a group. Using the contributions y_i^j from Table II, the cycle index of G is:

$$z_G = \frac{1}{4}(y_1^6 + y_1^2 y_2^2 + 2y_2^3) \qquad (3)$$

Using Eq. 3 for the cycle index and Polya's theorem, we obtain for the configuration-counting series:

$$C = \frac{1}{4}\Big((x_1 + x_2)^6 (x_3 + x_4)^6 + (x_1 + x_2)^2 (x_3 + x_4)^2 (x_1^2 + x_2^2)^2 (x_3^2 + x_4^2)^2 + 2(x_1^2 + x_2^2)^3 (x_3^2 + x_4^2)^3 \Big)$$

The expansion of the above expression is done by the following function:

```
     ⍝FOR APL CONF 8/20/74
[2]   X1← 2 1 ρ 1 1
[3]   X2← 2 1 ρ 100 1
[4]   X3← 2 1 ρ 10000 1
[5]   X4← 2 1 ρ 1000000 1
[6]   X1X22←TEMP M TEMP←X1 A X2
[7]   X1X26←X1X22 M X1X22 M X1X22
[8]   X3X42←TEMP M TEMP←X3 A X4
[9]   X3X46←X3X42 M X3X42 M X3X42
[10]  X12←X1 M X1
[11]  X22←X2 M X2
[12]  X32←X3 M X3
[13]  X12X222←TEMP M TEMP←X12 A X22
[14]  X42←X4 M X4
[15]  X32X422←TEMP M TEMP←X32 A X42
[16]  X12X223←X12X222 M X12 A X22
[17]  X32X423←X32X422 M X32 A X42
[18]  P1←X1X26 M X3X46
[19]  P2←X1X22 M X3X42 M X12X222 M X32X422
[20]  P3←(K 2) M X12X223 M X32X423
[21]  Z←(K 0.25) M P1 A P2 A P3
    ∇
```

The exponents of the term $X_1^i X_2^j X_3^k X_4^l$ are represented by the power $1000000i + 10000j + 100k + l$. This trick minimizes cpu time and storage requirements and can be used when i, j, k, l will not exceed 99.

The final answer, Z, is formatted with the PRINT function (Appendix) and shown in Table III.

Given a weight vector from the configuration-counting series we can obtain the number of equivalence classes of configurations with that weight vector. For instance to find the number of configurations having 5 red objects, 1 blue object, 1 round object and 5 square objects (i.e., the number of configurations with weight vector (5, 1, 1, 5), we compute the coefficient of

$$x_1^5 x_2^1 x_3^1 x_4^5$$

in the configuration-counting series. It is seen to be equal to 10. The 10 configurations are shown in Figure 3 with the blue object being crosshatched.

OTHER EXAMPLES OF GENERATING FUNCTIONS USING SYMBOLIC DIVISION

Table IV contains some examples of generating functions that are useful in counting binary patterns. The coefficient of x^n in each case is obtained by the use of the APL symbolic functions for division A OVER B TO N. Here the Polynomial A is to be divided by B and the answer returned is to have N terms. The remainder is contained in REM. The function N COEFF MATRIX then selects the coefficient of the Nth power. If there is no Nth power, a zero is returned.



Fig. 3 — Configurations of a 2 by 3 rectangle with 6 objects, 5 of which are red, 1 is blue (shown crosshatched), 1 is round, and 5 are square.

Table III

Formatted Answer Z

```
      7 PRINT Z
```

6000600	6000501	6000402	6000303	6000204	6000105	6000006
1	2	6	6	6	2	1
5010600	5010501	5010402	5010303	5010204	5010105	5010006
2	10	24	32	24	10	2
4020600	4020501	4020402	4020303	4020204	4020105	4020006
6	24	63	78	63	24	6
3030600	3030501	3030402	3030303	3030204	3030105	3030006
6	32	78	104	78	32	6
2040600	2040501	2040402	2040303	2040204	2040105	2040006
6	24	63	78	63	24	6
1050600	1050501	1050402	1050303	1050204	1050105	1050006
2	10	24	32	24	10	2
60600	60501	60402	60303	60204	60105	60006
1	2	6	6	6	2	1

REFERENCES

[1] Kellerman, Rodgers, "APL Tools for Combinatorics," Proceedings of the Sixth International Users Conference, May 14-17, Anaheim, California, U.S.A.

[2] Kellerman, A., Kellerman, E., "Polynomial Manipulation in APL," IBM Technical Report No. TR 01.1445.

[3] Polya, "Kombinatorische Anzahlbestimmungen fur Gruppen, "Acta Mathematica, 68, May, 1937.

[4] Beckenbach (Editor), Applied Combinatorial Mathematics, Wiley and Sons, Inc., New York, 1964.

[5] Riordan, J., An Introduction to Combinatorial Analysis, John Wiley and Sons, Inc., New York, 1958.

Table IV

Examples of Generating Functions for Counting Binary Patterns

	Generating Function	Specific Case	APL Answer
N digit binary sequences that have the pattern 010 occurring at the Nth digit	$\frac{1-2x+x^2-x^3}{1-2x+x^2-2x^3}$	5 digit binary sequence	3
Number of N digit binary sequences that have pattern 010 occurring for the first time at the Nth digit	$\frac{x^3}{1-2x+x^2-x^3}$	4 digit binary sequence	6
Number of N digit binary sequences in which an occurrence of the pattern 010 is followed by an occurrence of the pattern 010	$\frac{x^6}{1-6x+13x^2-12x^3+4x^4+x^5-3x^6+2x^7}$	8 digit binary sequence	23

e.g., Details of the third case in the table:

```
          C                         D
  6                                   0   1   2   3   4   5   6   7
  1                                   1  ¯6  13 ¯12   4   1  ¯3   2

        C OVER D TO 5
  10                            9            8              7
            6
  201                          72           23              6
            1


           8 COEFF C OVER D TO 5
  23
```

APPENDIX

FUNCTION LISTINGS

```
      ⍝ A M B MULTIPLIES SYMBOLICALLY TWO POLYNOMIALS A AND B
      ∇M[⎕]∇
    ∇ C←A M B;I;P;N;T
      N←ρI←⍋P←,A[1;]∘.+B[1;]
      P←P[I]
      C←(2,ρT)ρT,,((T←(P≠(¯1↓1⌽P),P[ρP]+1)/P)∘.=P)+.×(N,1)ρ(,A[
      2;]∘.×B[2;])[I]
      C←(2,(⌈/1,(ρC)[2]))↑C←(C[2;]≠0)/C
    ∇
```

```
      ⍝ A A̲ B ADDS TWO POLYNOMIALS A AND B
      ∇A̲[⎕]∇
    ∇ C←A A̲ B;I;P;N;T
      N←ρI←⍋P←A[1;],B[1;]
      P←P[I]
      C←(2,ρT)ρT,,((T←(P≠(¯1↓1⌽P),P[ρP]+1)/P)∘.=P)+.×(N,1)ρ(A[
      2;],B[2;])[I]
      C←(2,(⌈/1,(ρC)[2]))↑C←(C[2;]≠0)/C
    ∇
```

```
      ⍝ K GENERATES A CONSTANT POLYNOMIAL  N
      ∇K[⎕]∇
    ∇ Z←K N
      Z← 2 1 ρ0,N
    ∇
```

```
      ⍝ B OVER A DIVIDES A INTO B
      ∇OVER[⎕]∇
    ∇ Z←B OVER A;Y;I
      Z← 2 1 ρ0
      Y←B
      I←0
     LP:→(N̲N̲<I←I+1)/OUT
      X←Y D̲ A
      Z←Z A̲ 2 1 ↑X
      Y← 0 1 ↓X
      →LP
     OUT:R̲E̲M̲←Y
    ∇
```

```
      ⍝ TO IS USED WITH OVER WHERE A OVER B TO N, N IS THE NUMBER OF TERMS
      ∇TO[⎕]∇
    ∇ Z←A TO N
      N̲N̲←N
      Z←A
    ∇
```

```
      ⍝B D̲ A WHERE A IS IN ASCENDING ORDER IS CALLED BY OVER FOR
      ⍝DIVISION
      ∇D̲[⎕]∇
    ∇ Z←B D̲ A;R;S
      Q← 2 1 ρ(B[1;1]-A[1;1]),B[2;1]÷A[2;1]
      S←A M Q
      S[2;]←-S[2;]
      R←B A̲ S
      Z←Q AND⌽R
    ∇
```

```
      ⍝P AT N EVALUATES POLYNOMIAL P FOR ARGUMENT N
      ∇AT[⎕]∇
    ∇ Z←P AT N
[1]   Z←+/P[2;]×N*P[1;]
    ∇

      ⍝ N COEFF MATRIX PICKS COEFFICIENT OF N POWER IN POLYNOMIAL,MATRIX
      ∇COEFF[⎕]∇
    ∇ Z←N COEFF MATRIX
[1]   Z←(Z≤(⍴MATRIX)[2])×Z←MATRIX[1;]⍳N
[2]   →0 IF Z=0
[3]   Z←MATRIX[2;Z]
    ∇


∇PRINT[⎕]∇
    ∇ K PRINT POL;N;I;LEN
[1]  ⍝ PRINTS K TERMS OF POL PER LINE USING NN FOR FIELD WIDTH
[2]   N←⌈(LEN←1↓⍴POL)÷K
[3]   I←0
[4]  LOOP:→LAST IF N=I←I+1
[5]   (NN,0)⍕(2,K)↑(0,K×I-1)↓POL
[6]   ' '
[7]   →LOOP
[8]  LAST:→LAST2 IF 0=K|LEN
[9]   (NN,0)⍕(-2,K|LEN)↑POL
[10]  →0
[11] LAST2:(NN,0)⍕(-2,K)↑POL
    ∇
```